```
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%                                                                               %
%      File:          ch2_koryak.dtr                                            %
%      Purpose:       koryak nouns                                              %
%      Author:        Dunstan Brown, December 20, 2011                          %
%      Email:         d.brown@surrey.ac.uk                                      %
%      Address:       SMG, University of Surrey, Guildford GU2 7XH              %
%      Documentation: Brown & Hippisley 2012 'Network Morphology',              %
%                     chapter 2, section 2.2.3                                  %
%      Related Files: ch2_koryak.dmp                                            %
%      Version:       1.00                                                      %
%                                                                               %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %

% The purpose of this fragment is to illustrate the proposed structure of   %
% of the Koryak noun paradigm given the fact that Declension I has          %
% neutralization of number outside the absolutive, and Declension II has    %
% dual-plural syncretism outside the absolutive.                            %
%                                                                           %
% A note on the correspondences between the representation in (16) and       %
% and the transcription used here:                                          %
%     N is a velar nasal                                                    %
%     @ is schwa                                                            %
%     TS represents the č in (16) in section 2.2.3.                         %
%                                                                           %
% The fragment also implements a rule of vowel harmony described by         %
% Žukova (1972: 27).                                                        %
%                                                                           %
% As we argue in our discussion of (16) in section 2.2.3,                   %
% underspecification is simultaneously combined with a referral, in the     %
% equation <mor du> == "<mor pl>". The syncretism of the ergative and        %
% locative results from the underspecification of the suffix _k, in         %
% <mor suffix> == _k.                                                       %
%                                                                           %
% According to Žukova (1972: 45) declensions I and II differ in that in the %
% latter 'articles' are affixed before the case marking in the              %
% non-absolutive forms. These articles distinguish singular and plural, so  %
% their absence in declension I means that there is no number               %
% distinction outside of the absolutive. We have treated this difference    %
% in terms of stems, so that <stem 3> is the stem with the singular         %
% article, and stem 4 is the stem with the plural article. For declension I %
% these stems are defined without the articles. <stem 1> is                 %
% the stem for the absolutive singular and <stem 2> is the stem for the     %
% absolutive dual and absolutive plural.                                    %

MOR_NOUN:
    <mor> ==
    <mor sg> == "<stem 3>" "<mor suffix>"
    <mor pl> == "<stem 4>" "<mor suffix>"
    <mor du> == "<mor pl>"
    <mor sg abs> == "<stem 1>"
    <mor pl abs> == "<stem 2>" _w
    <mor suffix> == _k
    <mor suffix abl> == _Nqo
    <mor suffix trans> == _jp@N
    <mor suffix dat> == _N
    <mor suffix adit> == _jt@N

% <mor suffix des> is either -nu or -no depending on the stem type (whether %
% this lowers the vowel or not). The vowel harmony is dealt with at the     %
% node VH. <mor suffix narr-caus> and <mor suffix cont> are similarly       %
% subject to the effects of vowel harmony.                                  %


    <mor suffix des> == _n VH:<u "<stem type>" >
    <mor suffix narr-caus> == _kj  VH:<i "<stem type>" > _t
    <mor suffix cont> == _j VH:<i "<stem type>" > _t VH:<e "<stem type>" >.


% While declension I and declension II differ, because of the use of the    %
% suffixed articles, they share the same singular and plural absolutive     %
% realizations (see MOR_NOUN), but differ in terms of their dual            %
% absolutive.                                                               %

DECL_1:
    <> == MOR_NOUN
    <mor du abs> == "<stem 2>" _t
    <mor suffix erg> == _te.

DECL_2:
    <> == MOR_NOUN
    <mor sg cont> == "<stem 2>" "<mor suffix cont>"
    <mor du abs> == "<stem 2>" _nte.
```

```
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%                                                                             %
%                       LEXEMIC HIERARCHY                                     %
%                                                                             %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %

% Žukova (1972: 45) says that declension I accounts for the majority of     %
% Koryak nouns, which is why we have made it the default declension (i.e.   %
% <mor> == DECL_1).                                                         %

NOUN:
    <> == UNDEF
    <syn cat> ==  n
    <mor> == DECL_1
    <stem> == STEM_1.


% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%                                                                             %
%                       STEMS                                                 %
%                                                                             %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %

% Differences in stems caused by vowel harmony are treated here. As en'pič  %
% contains series I vowels in its base form these will be subject to        %
% lowering in the presence of certain suffixes. This is treated by the      %
% node STEM_1 which lowers the vowels of the stem according to the suffix   %
% with which they are used. In contrast, appa inherits from STEM_2, for     %
% which no lowering is required, because the base form contains series I    %
% vowels. We are required, however, to model the alternation in the plural  %
% article according to the following case suffix (-j@ka- vs. -j@k).         %

% This is not an exhaustive treatment of Koryak stems and vowel harmony,    %
% but it is sufficient to show that we can model the shape                  %
% of the paradigm in the way proposed in section 2.2.3 and accommodate the  %
% the properties of the stems.                                              %

STEM:
    <stem> == "<stem final "<root final>" >"
    <stem final vowel> == "<root>"
    <stem final cons> == "<root>" "<root vowel>".


STEM_1:
    <> == STEM
    <stem 1> == "<root>"
    <stem 3> == "<stem v_harm "<mor suffix>" >"
    <stem 4> == "<stem v_harm "<mor suffix>" >"
    <stem v_harm _Nqo> == "<stem low>"
    <stem v_harm _N> == "<stem low>"
    <stem v_harm _jt@N> == "<stem low>"
    <stem v_harm _jp@N> == "<stem low>".

STEM_2:
    <> == STEM
    <stem form> == _ka
    <stem form _k> ==
    <stem form _Nqo> == _ka
    <stem form _N> == _k@
    <stem form _n o> == _TSge
    <stem form _kj e _t> ==
    <stem 3> == "<stem low>" _na
    <stem 4> == "<stem low>" _j@ "<stem form "<mor suffix>" >".


% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%                                                                             %
%                       VOWEL HARMONY                                         %
%                                                                             %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %

% According to Žukova (1972) there are two series of vowel                   %
%    Series I:  i  e  u                                                       %
%    Series II: e  a  o                                                       %
% When formatives are combined, the higher vowels (i.e. series I) will       %
% assimilate to the lower ones (i.e. series II).                             %
% This can be observed in the contrast between SG DES appa-na-no 'papa'      %
% and SG DES en'piči-nu, where -nu is lowered to -no in the former.          %
% In SG ABL we can see an'peče-ŋqo, where the vowels in en'piči- have been   %
% lowered before -ŋqo. As it is the higher vowels which harmonize with the   %
% the lower vowels, our implementation represents the relevant vowels at a   %
```

```
% node VH where the paths extended by low (<i low> etc.) are the lowered    %
% series II equivalents of series I.                                        %

VH:
    <i> == i
    <e> == e
    <u> == u
    <i low> == e
    <e low> == a
    <u low> == o.

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%                                                                           %
%                       NOUNS                                               %
%                                                                           %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %


Appa:
    <> == NOUN
    <mor> == DECL_2
    <root> == appa
    <root type> == low
    <root final> == vowel
    <stem> == STEM_2
    <gloss> == papa.

EnpiTS:
    <> == NOUN
    <root> == VH:<e> n'p <root vowel> _TS
    <root vowel> == VH:<i>
    <root final> == cons
    <root type> == not_low
    <gloss> == father.

# show
    <gloss>
    <mor sg abs>
    <mor sg loc>
    <mor sg erg>
    <mor sg abl>
    <mor sg trans>
    <mor sg dat>
    <mor sg dat>
    <mor sg adit>
    <mor sg des>
    <mor sg narr-caus>
    <mor sg cont>
    <mor pl abs>
    <mor pl loc>
    <mor pl erg>
    <mor pl abl>
    <mor pl trans>
    <mor pl dat>
    <mor pl dat>
    <mor pl adit>
    <mor pl des>
    <mor pl narr-caus>
    <mor pl cont>
    <mor du abs>
    <mor du loc>
    <mor du erg>
    <mor du abl>
    <mor du trans>
    <mor du dat>
    <mor du dat>
    <mor du adit>
    <mor du des>
    <mor du narr-caus>
    <mor du cont>.


# hide
    DECL_1 DECL_2 NOUN MOR_NOUN STEM STEM_1 STEM_2 VH.
```